Math 2650

A. J. Meir

Second Order Equations

A Brief Review

Consider a linear, homogeneous, constant coefficient second order equation

$$\frac{d^2 x}{dt^2} + a\,\frac{dx}{dt} + bx = 0.$$

The characteristic polynomial corresponding to this equation is

$$r^2 + ar + b = 0$$

whose roots are r_1 and r_2 where

$$r_1 = \frac{-a + \sqrt{a^2 - 4\,b}}{2} \qquad r_2 = \frac{-a - \sqrt{a^2 - 4\,b}}{2}\ .$$

We will consider three distinct cases (depending on the value of $a^2 - 4\,b$):

I. Two distinct real roots.
If $0 < a^2 - 4\,b$ there are two real distict roots $r_1 \neq r_2$ and the homogenous solution is

$$x_h(t) = c_1\,e^{r_1 t} + c_2\,e^{r_2 t}.$$

II. A double root.
If $a^2 - 4\,b = 0$ there are two equal roots $r = r_1 = r_2$ and the homogenous solution is

$$x_h(t) = c_1\,e^{r\,t} + c_2\,t\,e^{r\,t}.$$

III. Complex conjugate roots.
If $a^2 - 4\,b < 0$ there are two complex conjugate roots

$$r_1 = \alpha + I\beta \qquad r_2 = \alpha - I\beta$$

where $\alpha = -\frac{a}{2}$, $\beta = \frac{\sqrt{4\,b - a^2}}{2}$ and I is the square root of -1. The homogenous solution is

$$x_h(t) = c_1\,e^{\alpha t}\cos(\beta\,t) + c_2\,e^{\alpha t}\sin(\beta\,t).$$

Examples:

1. Consider the differential equation $\frac{d^2 x}{dt^2} - \frac{dx}{dt} - 6 = 0$. The characteristic polynomial is

$r^2 - r - 6 = 0$. The roots of the characteristic polynomial are

```
> solve(r^2-r-6=0);
```

$$3, -2$$

so the solution of the equation is $x_h(t) = c_1 e^{3t} + c_2 e^{-2t}$.

2. Consider the differential equation $\frac{d^2 x}{dt^2} + 6 \frac{dx}{dt} + 9 = 0$. The characteristic polynomial is $r^2 + 6r + 9 = 0$. The roots of the characteristic polynomial are

```
> solve(r^2+6*r+9=0);
```

$$-3, -3$$

so the solution of the equation is $x_h(t) = c_1 e^{-3t} + c_2 t e^{-3t}$.

3. Consider the differential equation $\frac{d^2 x}{dt^2} + 4 \frac{dx}{dt} + 5x = 0$. The characteristic polynomial is $r^2 + 4r + 5 = 0$. The roots of the characteristic polynomial are

```
> solve(r^2+4*r+5=0);
```

$$-2 + I, -2 - I$$

Note, Maple writes I for $\sqrt{-1}$, so the solution of the equation is $x_h(t) = e^{-2t} \left(c_1 \cos(t) + c_2 \sin(t)\right)$.

4. Consider the initial value problem $\frac{d^2 x}{dt^2} + 4x = 0, x(0) = 1, \frac{dx}{dt}(0)=1$. The characteristic polynomial is $r^2 + 4 = 0$. The roots of the characteristic polynomial are

```
> solve(r^2+4=0);
```

$$2 I, -2 I$$

so the general solution of the equation is $x_h(t) = c_1 \cos(2t) + c_2 \sin(2t)$. We now use the initial conditions to solve for c_1 and c_2.

```
> sol:=c[1]*cos(2*t)+c[2]*sin(2*t);
```

$$sol := c_1 \cos(2t) + c_2 \sin(2t)$$

```
> eq1:=subs(t=0,sol)=1;
```

$$eq1 := c_1 \cos(0) + c_2 \sin(0) = 1$$

```
> eq2:=subs(t=0,diff(sol,t))=1;
```

$$eq2 := -2 c_1 \sin(0) + 2 c_2 \cos(0) = 1$$

```
> solve({eq1,eq2},{c[1],c[2]});
```

$$\left\{c_1 = 1, c_2 = \frac{1}{2}\right\}$$

so the solution is $x(t) = \cos(2t) + \frac{\sin(2t)}{2}$.

Undetermined Coefficients

This method applies to special classes of nonhomogeneous second order equations. It is **crucial** that the homogeneous problem have constant coefficients.
Consider a nonhomogeneous constant coefficient second order equation

$$\frac{d^2x}{dt^2} + a\frac{dx}{dt} + bx = f(t).$$

If the right hand side $f(t)$ has the form (exponential times a polynomial times a trigonometric polynomial) we guess a particular solution of the same form
$f(t) = e^{kt}(a_n t^n + a_{n-1} + t^{n-1} + ... + a_0)(\cos(\omega t) + \sin(\omega t))$.
Then guess a particular solution of the form
$x_p(t) = e^{kt}(A_n t^n + A_{n-1} + t^{n-1} + ... + A_0)\cos(\omega t) + e^{kt}(B_n t^n + B_{n-1} + t^{n-1} + ... + B_0)\sin(\omega t)$.

If the above solution x_p is a solution of the homogeneous equation you need to multiply it by t^s (s counts the number of times x_p is a solution of the homogeneous problem, and for a second order equation s is either 1 or 2).

Example:

Consider the differential equation $\frac{d^2x}{dt^2} + 4\frac{dx}{dt} + 5x = \cos(t)$. The characteristic polynomial is $r^2 + 4r + 5 = 0$. The roots of the characteristic polynomial are

```
> solve(r^2+4*r+5=0);
```

$$-2 + I, -2 - I$$

so the solution of the homogeneous problem is $x_h(t) = e^{-2t}\left(c_1\cos(t) + c_2\sin(t)\right)$ we now have to find one particular solution of $\frac{d^2x}{dt^2} + 4\frac{dx}{dt} + 5x = \cos(t)$ in order to obtain the general solution.
The idea is to find this solution using an intelligent guess. The second derivative of a cosine term is again a cosine term, but the first one is a sine term. Therefore we try an expression $x_p(t) = A\cos(t) + B\sin(t)$ as initial guess. The goal is to find the undetermined coefficients A, B (in $x_p(t)$) in such a manner that x_p is the desired particular solution.

```
> restart:
> guess:=y(t)=A*cos(t)+B*sin(t);
```

$$guess := y(t) = A\cos(t) + B\sin(t)$$

```
> param:={A,B};
```

$$param := \{A, B\}$$

```
> eq:=diff(y(t),t$2)+4*diff(y(t),t)+5*y(t)=cos(t);
```

$$eq := \frac{d^2}{dt^2}y(t) + 4\left(\frac{d}{dt}y(t)\right) + 5y(t) = \cos(t)$$

```
> subs(guess,eq);
```

$$\frac{\partial^2}{\partial t^2}(A\cos(t) + B\sin(t)) + 4\left(\frac{\partial}{\partial t}(A\cos(t) + B\sin(t))\right) + 5A\cos(t) + 5B\sin(t)$$

$= \cos(t)$

```
> simplify(%);
```

$$4\,A\cos(t) + 4\,B\sin(t) - 4\,A\sin(t) + 4\,B\cos(t) = \cos(t)$$

Since $\cos(t)$ and $\sin(t)$ are linearly independent, the coefficient $4\,A + 4\,B$ of the cosine on the left hand side of the equation has to be equal to 1 (the coefficient of the cosine on the right), whereas the coefficient $4\,B - 4\,A$ of the sine has to be zero. This leads to two linear equations which we solve by

```
> sloveparam:=solve({4*A+4*B=1,4*B-4*A=0},param);
```

$$sloveparam := \left\{B = \frac{1}{8}, A = \frac{1}{8}\right\}$$

Thus, $x_p(t) = \dfrac{\cos(t) + \sin(t)}{8}$ is a particular solution. We can check this by using the odetest-command.

```
> odetest(y(t)=(cos(t)+sin(t))/8,D(D(y))(t)+4*D(y)(t)+5*y(t)=
  cos(t));
```

$$0$$

Therefore the general solution is given as $x_{gen}(t) = x_h(t) + x_p(t)$, i.e.

$x_{gen}(t) = e^{-2t}\left(c_1\cos(t) + c_2\sin(t)\right) + \dfrac{\cos(t) + \sin(t)}{8}$, a fact, which you can also test by inserting this expression into the d.e.:

```
> odetest(y(t)=exp(-2*t)*(c[1]*cos(t)+c[2]*sin(t))+(cos(t)+sin
  (t))/8,D(D(y))(t)+4*D(y)(t)+5*y(t)=cos(t));
```

$$0$$

If in addition we were given initial conditions, we would now use those to solve for the constants c_1 and c_2.

Exercise

Find the general solution of $\dfrac{d^2 x}{dt^2} - 3\,\dfrac{dx}{dt} - 4\,x = 2\sin(t)$.

```
> restart;
```

Exercise

Find the general solution of $\dfrac{d^2 x}{dt^2} + x(t) = 2\sin(t)$.

```
> restart;
```

A Trick

What went wrong? Clearly, $A\cos(t) + B\sin(t)$ solves the homogeneous problem.

```
> odetest(y(t)=A*cos(t)+B*sin(t),D(D(y))(t)+y(t)=0);
```

$$0$$

Try the guess $x_p(t) = t\,(A\cos(t) + B\sin(t))$ and see what you get:

Let us explore this last guess somewhat more. Consider the nonhomogeneous problem

$$\frac{d^2 x}{dt^2} + p\,\frac{dx}{dt} + q\,x(t) = a\cos(\beta\,t) + b\sin(\beta\,t)$$

where p and q are real numbers, $\beta > 0$ and $0 < a^2 + b^2$. We ask for conditions under which the guess $y(t) := t\,(A\cos(\beta\,t) + B\sin(\beta\,t))$ works. To this end we insert y into the differential equation.

```
> restart;
> y(t):=t*(A*cos(beta*t)+B*sin(beta*t));
```

$$y(t) := t\,(A\cos(\beta\,t) + B\sin(\beta\,t))$$

```
> diff(y(t),t$2)+p*diff(y(t),t)+q*y(t)=a*cos(beta*t)+b*sin
  (beta*t);
```

$$-2A\sin(\beta\,t)\,\beta + 2B\cos(\beta\,t)\,\beta + t\,(-A\cos(\beta\,t)\,\beta^2 - B\sin(\beta\,t)\,\beta^2) + p\,(A\cos(\beta\,t) + B\sin(\beta\,t) + t\,(-A\sin(\beta\,t)\,\beta + B\cos(\beta\,t)\,\beta)) + q\,t\,(A\cos(\beta\,t) + B\sin(\beta\,t)) = a\cos(\beta\,t) + b\sin(\beta\,t)$$

This yields the following four conditions for p, q, A and B:

$t\cos(\beta\,t)$: $\quad -A\,\beta^2 + p\,B\,\beta + q\,A = 0,$

$t\sin(\beta\,t)$: $\quad -B\,\beta^2 - p\,A\,\beta + q\,B = 0,$

$\cos(\beta\,t)$: $\quad 2\,B\,\beta + p\,A = a,$

$\sin(\beta\,t)$: $\quad -2\,A\,\beta + p\,B = b.$

```
> solve({-A*beta^2+p*B*beta+q*A=0,
  -B*beta^2-p*A*beta+q*B=0,
  2*B*beta+p*A=a,
  -2*A*beta+p*B=b},{q,p,B,A});
```

$$\left\{p = 0,\ B = \frac{1}{2}\,\frac{a}{\beta},\ q = \beta^2,\ A = -\frac{1}{2}\,\frac{b}{\beta}\right\}$$

Note that $a\cos(\beta\,t) + b\sin(\beta\,t)$ solves $\frac{d^2 x}{dt^2} + \beta^2\,x(t) = 0$ as the following shows

```
> with(DEtools):
> odetest(z(t)=a*cos(beta*t)+b*sin(beta*t),D(D(z))(t)+beta^2*z
  (t)=0);
```

$$0$$

consequently, one concludes that **the guess** $t\,(A\cos(\beta\,t) + B\sin(\beta\,t))$ **only works in case that** $A\cos(\beta\,t) + B\sin(\beta\,t)$ **solves the homogeneous problem.**

The Resonance Case

Consider a second order linear differential equation

$$\frac{d^2 x}{dt^2} + a\,\frac{dx}{dt} + b\,x = f(t)$$

where a, b are real numbers, and $f(t)$ has the form $f(t) = p_n(t)\,e^{\alpha t}\cos(\beta\,t) + q_n(t)\,e^{\alpha t}\sin(\beta\,t)$ with α, β real numbers and p_n, q_n polynomials in t of degree less than or equal to n.

Step 1. Set up standard trial function.

Step 2. Check whether any term solves the homogeneous d.e.

Step 3. If so, multiply by t and go to Step 2.
Step 4. If not, determine coefficients by inserting the improved trial function and its derivatives into the de.

Guesses for Other Forcing Functions

Find a particular solution for the following differential equations.

a)

$$\frac{d^2 x}{dt^2} - \frac{dx}{dt} - 6\,x(t) = 10\,\mathrm{e}^{2\,t}$$

```
> restart;
```

b)

$$\frac{d^2 x}{dt^2} - 3\,\frac{dx}{dt} = 64\,(t^3 - t^2)$$

```
> restart;
```

c)

$$\frac{d^2 x}{dt^2} + 4\,x(t) = 2\,t^2 \sin(2\,t)$$

```
> restart;
>
```